







TCAD: Process and Device Simulation

**2- μm N-well CMOS Process
TSUPREM-4 Results**




<i>Pattern</i>					
	NWELL	ACTIVE	POLY1	METAL	CONTACT
<i>Field</i>	dark	clear	clear	clear	dark

0. Layout:


```

$driver tmalayout
$step load
load name=VWF:0_0.t12 t12
$step cutline
cutline x1=0 y1=0 x2=18 y2=0
scale all xscale=1 yscale=1
save t11 name=layout:0_0.t11
    
```




1. Starting wafer:

```
$step load_mask
mask in.file=layout:0_0.tl1
$step init_mesh
method dy.oxide=0.1 err.fac=3
mesh grid.fac=3 ly.surf=0.5
  dy.surf=0.1 ly.activ=2
  dy.activ=0.2 ly.bot=10
  dy.bot=0.5 dx.min=0.2
initialize <100> boron=10
  resistiv
$step save_struct
savefile
  out.file=init_proc:0_0.tif
  tif
```




2. Initial oxidation:

```
$step init_oxide
diffusion time=25 temp=750
  t.final=1000 f.N2=3 f.O2=0.05
diffusion time=5 temp=1000
  dryO2
diffusion time=50 temp=1000
  f.H2=3 f.O2=1.7
diffusion time=5 temp=1000
  dryO2
diffusion time=25 temp=1000
  t.final=750 inert
$step extr_tox_init
extract oxide thicknes x=9
  prefix="tox_init "
  suffix="(0.38) um"
  out.file=init_proc:0_0.ext
```




3. N-well photolithography:

```
$step load_mask  
mask in.file=layout:0_0.tl1  
$step nwell_mask  
deposit photo thick=1  
expose mask=NWELL  
develop
```



4. Oxide and photoresist etch:

```
$step etch_oxide  
etch oxide  
$step strip_photo  
etch photo all
```




5. N-well implant oxidation:

```


$step implant_oxide
diffusion time=20 temp=750
  t.final=1000 f.N2=3 f.O2=0.05
diffusion time=5 temp=1000
  dryO2
diffusion time=8 temp=1000
  f.H2=3 f.O2=1.7
diffusion time=5 temp=1000
  dryO2
diffusion time=20 temp=1000
  t.final=750 inert
$step extr_tox_impl
extract oxide thicknes x=4
  prefix="toximps "
  suffix="(?) um"
  out.file=nwell:0_0.ext
extract oxide thicknes x=14
  prefix="toximpw "
  suffix="(0.1) um"
  out.file=nwell:0_0.ext

```



6. N-well implant.

```
$step well_implant
implant phos dose=5e12
  energy=150
$step extr_Npeak
select z=phosphorus
extract silicon val.extr x=14
  maximum prefix="Pmax_well "
  suffix=" cm**-3"
  out.file=nwell:0_0.ext
$step extr_Nsurf
select z=phosphorus
extract silicon val.extr x=14
  distance=0 prefix="Psur_well
  " suffix=" cm**-3"
  out.file=nwell:0_0.ext
```




7. N-well drive-in:

```


$step nwell_drive
diffusion time=60 temp=750
  t.final=1150 f.N2=3 f.O2=0.05
diffusion time=260 temp=1150
  f.N2=3 f.O2=3.5
diffusion time=60 temp=1150
  inert
diffusion time=60 temp=1150
  t.final=750 inert
$step extr_tox_well
extract oxide thicknes x=4
  prefix="tox_subs "
  suffix="(?) um"
  out.file=nwell:0_0.ext
extract oxide thicknes x=14
  prefix="tox_well "
  suffix="(0.3) um"
  out.file=nwell:0_0.ext
$step extr_xj_well
select z=doping
extract silicon d.extr x=14
  value=0 prefix="xj_well "
  suffix="(3.4) um"
  out.file=nwell:0_0.ext

```




8. Oxide etch:

```
$step strip_oxide
etch oxide all
$step extr_Npeak
select z=phosphorus
extract silicon val.extr x=14
  maximum prefix="Pmax_wella "
  suffix=" cm**-3"
  out.file=nwell:0_0.ext
$step extr_Nsurf
select z=phosphorus
extract silicon val.extr x=14
  distance=0 prefix="Psur_wella
  " suffix=" cm**-3"
  out.file=nwell:0_0.ext
```




9. Pad oxidation:

```
$step pad_oxide
diffusion time=20 temp=750
  t.final=950 f.N2=3 f.O2=0.05
diffusion time=60 temp=950
  dryO2
diffusion time=20 temp=950
  t.final=750 inert
$step extr_tox_pad
extract oxide thicknes x=9
  prefix="tox_pad "
  suffix="(0.03) um"
  out.file=field_impl:0_0.ext
```




10. Nitride deposition:

```
$step deposit_nitride  
deposit nitride thicknes=0.33
```




11. Active photolithography:

```
$step load_mask  
mask in.file=layout:0_0.tl1  
$step active_mask  
deposit photo thick=1  
expose mask=ACTIVE  
develop
```




12. Plasma nitride etch:

```
$step etch_nitride
etch nitride
```




13. Field implant photolithography.

```
$step pfield_mask  
deposit negative photo thick=1  
expose mask=NWELL  
develop
```




14. Field implant.

```
$step field_implant  
implant boron dose=1.5e13  
energy=70
```




15. Photoresist removal:

```
$step strip_photo
etch photo all
$step etch_nitride
etch nitride thicknes=0.2
$step extr_Npeak
select z=boron
extract silicon val.extr x=9
  maximum prefix="Bmax_fld "
  suffix=" cm**-3"
  out.file=field_impl:0_0.ext
$step extr_Nsurf
select z=boron
extract silicon val.extr x=9
  distance=0 prefix="Bsur_fld "
  suffix=" cm**-3"
  out.file=field_impl:0_0.ext
```




16. Field (LOCOS) oxidation:

```
$step field_oxide  
diffusion time=20 temp=750  
  t.final=950 f.N2=3 f.O2=0.05  
diffusion time=5 temp=950 dryO2  
diffusion time=125 temp=1000  
  f.H2=3 f.O2=1.7  
diffusion time=5 temp=950 dryO2  
diffusion time=20 temp=950  
  t.final=750 inert
```




17. Nitride and pad oxide etch:

```
$step strip_nitride
etch nitride all
$step extr_tox_fld
extract oxide thicknes x=9
  prefix="tox_fld "
  suffix="(0.65) um"
  out.file=field_ox:0_0.ext
extract oxide thicknes x=4
  prefix="tox_padn "
  suffix="(?) um"
  out.file=field_ox:0_0.ext
extract oxide thicknes x=14
  prefix="tox_padp "
  suffix="(?) um"
  out.file=field_ox:0_0.ext
$step etch_oxide
etch oxide thicknes=0.032
```



18. Sacrificial oxidation:

```
$step sacr_oxide
diffusion time=20 temp=750
  t.final=950 f.N2=3 f.O2=0.05
diffusion time=30 temp=950
  dryO2
diffusion time=20 temp=950
  t.final=750 inert
$step extr_tox_sac
extract oxide thicknes x=4
  prefix="tox_sacn "
  suffix="(0.02) um"
  out.file=vt_impl:0_0.ext
extract oxide thicknes x=14
  prefix="tox_sacp "
  suffix="(0.02) um"
  out.file=vt_impl:0_0.ext
```




19. Threshold implant.

```


$step implant
implant boron dose=2e+12
  energy=30
$step extr_Npeak
select z=boron
extract silicon val.extr x=4
  maximum prefix="Bmax_act "
  suffix=" cm**-3"
  out.file=vt_impl:0_0.ext
extract silicon val.extr x=9
  maximum prefix="Bmax_fld "
  suffix=" cm**-3"
  out.file=vt_impl:0_0.ext
$step extr_Nsurf
select z=boron
extract silicon val.extr x=4
  distance=0 prefix="Bsur_act "
  suffix=" cm**-3"
  out.file=vt_impl:0_0.ext
extract silicon val.extr x=9
  distance=0 prefix="Bsur_fld "
  suffix=" cm**-3"
  out.file=vt_impl:0_0.ext

```




20. Sacrificial oxide etch:

```
$step etch_oxide  
etch oxide thicknes=0.022
```




21. Gate oxidation:

```
$step gate_oxide
diffusion time=20 temp=750
  t.final=950 f.N2=3 f.O2=0.05
diffusion time=60 temp=950
  dryO2
diffusion time=20 temp=950
  inert
diffusion time=20 temp=950
  t.final=750 inert
$step extr_tox_gate
extract oxide thicknes x=4
  prefix="tox_gn "
  suffix="(0.03) um"
  out.file=gate_ox:0_0.ext
extract oxide thicknes x=14
  prefix="tox_gp "
  suffix="(0.03) um"
  out.file=gate_ox:0_0.ext
```




22. Poly-Si deposition:

```
$step deposit_poly1  
deposit poly thick=0.45  
spaces=2 phos=1e20 concen  
temp=610
```




23. Gate definition:

```
$step load_mask  
mask in.file=layout:0_0.tl1  
$step gate_mask  
deposit photo thick=1  
expose mask=POLY1  
develop
```




24. Plasma polysilicon etch:

```
$step etch_poly1  
etch poly  
$step strip_photo  
etch photo all
```



25. N+ S/D photolithography:

```
$step load_mask  
mask in.file=layout:0_0.tl1  
$step nsd_mask  
deposit negative photo thick=1  
expose mask=NWELL  
develop
```




26. N+ S/D implant.

```


$step N+ s/d_implant
implant arsenic dose=5e+15
energy=130 gaussian
$step extr_Npeak
select z=arsenic
extract silicon val.extr x=4
  maximum prefix="Amax_nsdi "
  suffix=" cm**-3"
  out.file=nsd_impl:0_0.ext
$step extr_Nsurf
select z=arsenic
extract silicon val.extr x=4
  distance=0 prefix="Asur_nsdi
  " suffix=" cm**-3"
  out.file=nsd_impl:0_0.ext
$step extr_xj_nsd
select z=doping
extract silicon d.extr x=4
  value=0 prefix="xj_nsdi "
  suffix="(0.2) um"
  out.file=nsd_impl:0_0.ext

```



27. Photoresist removal:

```
$step strip_photo  
etch photo all
```




28. N+ anneal:

```


$step N+ s/d_anneal
diffusion time=20 temp=750
  t.final=950 inert
diffusion time=60 temp=950
  inert
diffusion time=20 temp=950
  t.final=750 inert
$step extr_Npeak
select z=arsenic
extract silicon val.extr x=4
  maximum prefix="Amax_nsd "
  suffix=" cm**-3"
  out.file=nsd_impl:0_0.ext
$step extr_Nsurf
select z=arsenic
extract silicon val.extr x=4
  distance=0 prefix="Asur_nsd "
  suffix=" cm**-3"
  out.file=nsd_impl:0_0.ext
$step extr_xj_nsd
select z=doping
extract silicon d.extr x=4
  value=0 prefix="xj_nsd "
  suffix="(0.35) um"
  out.file=nsd_impl:0_0.ext

```



29. P+ S/D photolithography:

```
$step load_mask  
mask in.file=layout:0_0.tl1  
$step psd_mask  
deposit photo thick=1  
expose mask=NWELL  
develop
```




30. P+ S/D implant.

```


$step P+ s/d_implant
implant boron dose=5e+15
energy=40
$step extr_Npeak
select z=boron
extract silicon val.extr x=14
  maximum prefix="Bmax_psd "
  suffix=" cm**-3"
  out.file=psd_impl:0_0.ext
$step extr_Nsurf
select z=boron
extract silicon val.extr x=14
  distance=0 prefix="Bsur_psd "
  suffix=" cm**-3"
  out.file=psd_impl:0_0.ext
$step extr_xj_psd
select z=doping
extract silicon d.extr x=14
  value=0 prefix="xj_psd "
  suffix="(0.55) um"
  out.file=psd_impl:0_0.ext

```



31. Photoresist removal:

```
$step strip_photo  
etch photo all
```




32. PSG deposition and densification:

```


$step deposit_psg
deposition oxide thicknes=0.7
  spaces=4 phos=1e20 concen
$step densify_psg
diffusion time=10 temp=750
  t.final=950 inert
diffusion time=5 temp=950 dryO2
diffusion time=20 temp=950 f.H2=3
  f.O2=1.7
diffusion time=5 temp=950 dryO2
diffusion time=10 temp=950
  t.final=750 inert
$step extr_tox_psg
extract oxide thicknes x=9
  prefix="tox_psgf " suffix="(?) um"
  out.file=psg_dep:0_0.ext
extract oxide thicknes x=4
  prefix="tox_psgn " suffix="(?) um"
  out.file=psg_dep:0_0.ext
extract oxide thicknes x=14
  prefix="tox_psgp " suffix="(?) um"
  out.file=psg_dep:0_0.ext
$step extr_xj
select z=doping
extract silicon d.extr x=4 value=0
  prefix="xj_nch " suffix="(0.35)
  um" out.file=psg_dep:0_0.ext
extract silicon d.extr x=14 value=0
  prefix="xj_pch " suffix="(0.6) um"
  out.file=psg_dep:0_0.ext

```




33. Contact photolithography:

```
$step load_mask  
mask in.file=layout:0_0.tl1  
$step contact_mask  
deposit photo thick=1  
expose mask=CONTACT  
develop
```




34. Contact etch:

```
$step etch_contact  
etch oxide  
$step strip_photo  
etch photo all
```




35. Metallization:

```
$step deposit_al  
deposit aluminum thick=0.6  
spaces=3
```



36. Metal photolithography.

```
$step load_mask  
mask in.file=layout:0_0.tl1  
$step metal_mask  
deposit photo thick=1  
expose mask=METAL  
develop
```




37. Metal etch:

```


$step etch_metal
etch aluminum
$step strip_photo
etch photo all
$step print_layers
select z=boron
print.1d x.val=4 layers
select z=doping
print.1d x.val=4 layers
$step extr_Npeak
select z=boron
extract silicon val.extr x=4
    maximum prefix="Bmax_nch "
    suffix=" cm**-3"
    out.file=metal:0_0.ext
$step extr_Nsurf
select z=boron
extract silicon val.extr x=4
    distance=0 prefix="Bsur_nch "
    suffix=" cm**-3"
    out.file=metal:0_0.ext

```



38. Truncate/reflect Nmos:

```
$step truncate_nmos
structure truncate x=8 right
$step reflect_nmos
structure reflect left
$step def_electrode
electrode name=Source x=-4 y=0
electrode name=Gate x=0 y=0
electrode name=Drain x=4 y=0
electrode name=Subs bottom
$step save_nmos
savefile
  out.file=nmos_md:0_0.md
  medici poly.ele elec.bot
```



39. Truncate/reflect Pmos:

```

$step truncate_pmos
structure truncate x=11 left
$step reflect_pmos
structure reflect right
$step truncate_bottom
structure truncate y=3 bottom
$step def_electrode
electrode name=Source x=22 y=0
electrode name=Gate x=18 y=0
electrode name=Drain x=14 y=0
electrode name=Subs bottom
$step save_pmos
savefile
  out.file=pmos_md:0_0.md
  medici poly.ele elec.bot

```

